微分(Diff)

解説

単純な微分(diff)

単純な一変数関数の一次微分は，以下の通り．

```
> diff(x^2-3*x+2,x);   #res: 2x-3
```

高次の微分は，微分変数を必要なだけ並べる．

```
> diff(sin(x),x,x);    #res: -sin(x)
```

さらに高次では次のような記法が便利．これはxについての3次微分を表わす．

```
> diff(x^4,x$3);       #res: 24x
```

偏微分

複数の変数を持つ関数の偏微分も同様にして記述できる．

```
> eq1:=(x+y)/(x*y);diff(eq1,x);
> diff(eq1,y);
```

$$eq1 := \frac{x+y}{x\,y}$$

$$\frac{1}{x\,y} - \frac{x+y}{x^2\,y}$$

$$\frac{1}{x\,y} - \frac{x+y}{x\,y^2} \quad \textbf{(1.1.2.1)}$$

級数展開(series)

Taylor級数は以下のようにして，中心点(x=a)，次数(4次)を指定する．

```
> t1:=series(sin(x),x=a,4);
```

$$t1 := \sin(3) + \cos(3)\,(x-3) - \frac{1}{2}\sin(3)\,(x-3)^2 - \frac{1}{6}\cos(3)\,(x-3)^3 + \mathrm{O}\left((x-3)^4\right) \quad \textbf{(1.1.3.1)}$$

このままでは関数定義することができないので，convertを使って多項式(polynomial)に変換する．

```
> e1:=convert(t1,polynom);
>  f1:=unapply(e1,x);
```

$$e1 := \sin(3) + \cos(3)\,(x-3) - \frac{1}{2}\sin(3)\,(x-3)^2 - \frac{1}{6}\cos(3)\,(x-3)^3$$

$$f1 := x \rightarrow \sin(3) + \cos(3)\,(x-3) - \frac{1}{2}\sin(3)\,(x-3)^2 - \frac{1}{6}\cos(3)\,(x-3)^3 \quad \textbf{(1.1.3.2)}$$

おまけ

以下のようにすると表示がきれい．

```
>  f:=unapply(x^4*exp(-y^2),(x,y));d:=Diff(f(x,y),x);
>  d=value(d);
```

$$f := (x,y) \rightarrow x^4\,\mathrm{e}^{-y^2}$$

$$d := \frac{\partial}{\partial x}\left(x^4\,\mathrm{e}^{-y^2}\right)$$

$$\frac{\partial}{\partial x}\left(x^4\,\mathrm{e}^{-y^2}\right) = 4\,x^3\,\mathrm{e}^{-y^2} \quad \textbf{(1.1.4.1)}$$

課題

1. 次の関数を微分せよ．

i) $x\log x$,　ii) $\dfrac{1}{(1+x)^3}$,　iii) $\sqrt{4\,x+3}$,　iv) $\dfrac{1}{a^2+\left(x-x_0\right)^2}$

2. 次の関数の1次から5次導関数を求めよ．

i) $\sin^2 x$,　ii) e^x

3. 次の関数とその1次導関数を同時にプロットし概形を確認し，さらに増減表を求めよ．

$$\frac{x}{x^2-2\,x+4}$$

4.次の関数の$x=3$での接線を求め，2つの関数を同時にプロットせよ．

$y=x^3-2\,x^2-35\,x$

5.以下の関数をx0まわりで３次までテイラー展開し，得られた関数ともとの関数をプロットせよ．さらに5次まで展開した場合はどう変化するか．

i) y=sin(x), x0=0,　ii) y=cos(x), x0=Pi/2

6.(発展課題）$f(x,y)=e^x\log(1+y)$を$x=0,\,y=0$のまわりで3次まで展開せよ．

解答例

1.

```
> diff(x*log(x),x);
  diff(1/(1 + x)^3,x);
  diff(sqrt(4*x + 3),x);
  diff(1/(a^2+(x-x0)^2),x);
```

$$\ln(x)+1$$

$$-\frac{3}{(1+x)^4}$$

$$\frac{2}{\sqrt{4x+3}}$$

$$-\frac{2x-2x0}{\left(a^2+(x-x0)^2\right)^2} \quad \textbf{(1.3.1.1)}$$

2.

```
> diff(sin(x)^2,x);
> diff(sin(x)^2,x$2);
```

$$2\sin(x)\cos(x)$$

$$2\cos(x)^2-2\sin(x)^2 \quad \textbf{(1.3.2.1)}$$

```
> diff(exp(x),x);
```

$$e^x \quad \textbf{(1.3.2.2)}$$

以下略

3.

```
> f0:=unapply(x/(x^2-2*x+4),x):
> df:=unapply(diff(f0(x),x),x);
> plot([f0(x),df(x)],x);
```

$$df := x \to \frac{1}{x^2-2x+4}-\frac{x(2x-2)}{\left(x^2-2x+4\right)^2}$$



4.

与関数をf0と定義.

```
> f0:=unapply(x^3  - 2*x^2  - 35*x,x);
```

$$f0 := x \to x^3-2x^2-35x \quad \textbf{(1.3.4.1)}$$

微分関数をdfと定義

```
> df:=unapply(diff(f0(x),x),x);
```

$$df := x \to 3x^2-4x-35 \quad \textbf{(1.3.4.2)}$$

接点(x0,f0(x0))で傾きdf(x0)の直線をf1と定義.

```
> x0:=3;
> eq1:=df(x0)*(x-x0)+f0(x0);
> f1:=unapply(eq1,x);
```

$$x0 := 3$$

$$eq1 := -20x-36$$

$$f1 := x \to -20x-36 \quad \textbf{(1.3.4.3)}$$

2つの関数を同時にプロット.

```
> plot([f0(x),f1(x)],x=-5..5);
```



5. i)

先ず与関数をf0と定義

```
> f0:=unapply(sin(x),x);
```

$$f0 := x \to \sin(x) \quad \textbf{(1.3.5.1)}$$

テイラー展開した結果をeq1とする．関数として定義するためにeq1を多項式に変換し(convert), unapplyをかける.

```
> eq1:=series(f0(x),x=0,3);
> f1:=unapply(convert(eq1,polynom),x);
```

$$eq1 := x+\mathrm{O}\left(x^3\right)$$

$$f1 := x \to x \quad \textbf{(1.3.5.2)}$$

５次についても同様

```
> eq2:=series(f0(x),x=0,5);
> f2:=unapply(convert(eq2,polynom),x);
```

$$eq2 := x-\frac{1}{6}x^3+\mathrm{O}\left(x^5\right)$$

$$f2 := x \to x-\frac{1}{6}x^3 \quad \textbf{(1.3.5.3)}$$

3つの関数を同時プロット

```
> plot([f0(x),f1(x),f2(x)],x=-Pi..Pi);
```



5. ii)

```
> series(f0(x),x=Pi/2,3)
```

以外は前問とおなじ．

6.

```
> f:=unapply(exp(x)*log(1+y),(x,y));
```

$$f := (x,y) \to e^{x} \ln(y+1) \quad (1.3.7.1)$$

```
> eq1:=series(series(f(x,y),x=0,3),y=0,3);
```

$$eq1 := O(x^3) + \left(1 + \frac{1}{2}x^2 + x\right)y + \left(-\frac{1}{2}x - \frac{1}{2} - \frac{1}{4}x^2\right)y^2 + O(y^3) \quad (1.3.7.2)$$

```
> g:=unapply(convert(convert(eq1,polynom),polynom),(x,y));
```

$$g := (x,y) \to \left(1 + \frac{1}{2}x^2 + x\right)y + \left(-\frac{1}{2}x - \frac{1}{2} - \frac{1}{4}x^2\right)y^2 \quad (1.3.7.3)$$

```
> plot3d([f(x,y),g(x,y)],x=-1..1,y=-1..1,axes=box);
```



全微分

全微分を計算するときは，Dを用いる．

```
>  f:=unapply(x^4*exp(-y^2),(x,y));
  D(f(x,y));
  (D@@2)(f(x,y));
```

$$f := (x,y) \to x^4 e^{-y^2}$$

$$4\,D(x)\,x^3 e^{-y^2} + x^4 D\left(e^{-y^2}\right)$$

$$4\,D^{(2)}(x)\,x^3 e^{-y^2} + 12\,D(x)^2 x^2 e^{-y^2} + 8\,D(x)\,x^3 D\left(e^{-y^2}\right) + x^4 D^{(2)}\left(e^{-y^2}\right) \quad (1.3.8.1)$$

ここで，D(x)などはxの全微分を表わす．これは，x,yを変数としているので

```
> diff(x,x);
  diff(exp(-y^2),y);
```

$$1$$

$$-2\,y\,e^{-y^2} \quad (1.3.8.2)$$

であるがMapleには分からない．そこで全微分の最終形を得るには，あらかじめD(x)などの結果を求めておき，subsで明示的に代入する必要がある．

```
> dd:=D(f(x,y)):
  eqs:={D(x)=diff(x,x),D(exp(-y^2))=diff(exp(-y^2),y)};
  subs(eqs,dd);
```

$$eqs := \left\{D(x) = 1,\ D\left(e^{-y^2}\right) = -2\,y\,e^{-y^2}\right\}$$

$$4\,x^3 e^{-y^2} - 2\,x^4 y\,e^{-y^2} \quad (1.3.8.3)$$

複合関数の微分

```
> diff(f(x)*g(x),x);
  diff(f(g(x)),x);
```

$$\left(\frac{d}{dx} f(x)\right) g(x) + f(x)\left(\frac{d}{dx} g(x)\right)$$

$$D(f)\,(g(x))\left(\frac{d}{dx} g(x)\right) \quad (1.3.9.1)$$

```
> f:=x->exp(x);
  g:=x->cos(x);
  diff(f(x)*g(x),x);
  diff(f(g(x)),x);
```

$$f := x \to e^{x}$$

$$g := x \to \cos(x)$$

$$e^{x}\cos(x) - e^{x}\sin(x)$$

$$-\sin(x)\,e^{\cos(x)} \quad (1.3.9.2)$$

積分(Int)

解説

単純な積分(int)

単純な不定積分.

```
> int(ln(x),x);          #res: x ln(x) - x
```

定積分を実行するには，積分変数の範囲を指定する.

```
> int(sin(x),x=-Pi..0);          #res: -2
```

特異点をもつ場合にも適切に積分結果を求めてくれる.

```
> int(1/sqrt(x*(2-x)),x=0..2);
```

$$\pi \tag{2.1.1.1}$$

無限区間(infinity)における定積分も同様に計算してくれる.

```
> int(1/(x^2+4),x=-infinity..infinity);
```

$$\frac{1}{2}\pi \tag{2.1.1.2}$$

部分積分法や置換積分法を用いる必要のある複雑な積分も一発で求まる.

```
> eq:=sqrt(4-x^2);int(eq,x);
```

$$eq := \sqrt{4-x^2}$$

$$\frac{1}{2}x\sqrt{4-x^2} + 2\arcsin\left(\frac{1}{2}x\right) \tag{2.1.1.3}$$

数学の公式集に載っているような積分も同じコマンドで求まる.

```
> eq2:=exp(-x^2);int(eq2,x=0..zz);
```

$$eq2 := \mathrm{e}^{-x^2}$$

$$\frac{1}{2}\sqrt{\pi}\,\mathrm{erf}(zz) \tag{2.1.1.4}$$

studentパッケージいろいろ

ちょっとぐらい難しい積分も，Mapleは単純にintコマンドだけで実行してくれる.しかし，時には，途中の計算法である部分積分，置換積分，部分分数展開が必要になる.このような計算はstudentパッケージに用意さている.

```
> with(student):
```

部分積分(integration by parts)

```
> intparts(Int(x*exp(x),x),x);
```

$$x\,\mathrm{e}^x - \left(\int \mathrm{e}^x\,\mathrm{d}x\right) \tag{2.1.2.1}$$

置換(change of variables)による積分

```
> Int((cos(x)+1)^3*sin(x), x);
>  changevar(cos(x)+1=u, Int((cos(x)+1)^3*sin(x), x=a..b),
  u);
>  changevar(cos(x)+1=u, int((cos(x)+1)^3*sin(x), x), u);
```

$$\int (\cos(x)+1)^3\sin(x)\,\mathrm{d}x$$

$$\int_{\cos(3)+1}^{\cos(4)+1} (-u^3)\,\mathrm{d}u$$

$$-\frac{1}{4}u^4 \tag{2.1.2.2}$$

部分分数(partial fraction)展開による積分では，convertコマンドを用いる.

```
> pf1:=convert(1/(1+x^3),parfrac,x);int(pf1,x);
```

$$pf1 := \frac{1}{3}\,\frac{2-x}{x^2-x+1} + \frac{1}{3(x+1)}$$

$$-\frac{1}{6}\ln(x^2-x+1) + \frac{1}{3}\sqrt{3}\arctan\left(\frac{1}{3}(2x-1)\sqrt{3}\right) + \frac{1}{3}\ln(x+1) \tag{2.1.2.3}$$

課題

1. 次の不定積分を求めよ.

i) $\int 4x+3\,\mathrm{d}x$, ii) $\int \frac{1}{1+e^x}\,\mathrm{d}x$, iii) $\int \frac{1}{e^{-x}+e^x}\,\mathrm{d}x$, iv) $\int \sqrt{1-x^2}\,\mathrm{d}x$

2. 次の定積分を求めよ.

i) $\int_0^{\pi} \sin x\,\mathrm{d}x$, ii) $\int_0^1 \arctan x\,\mathrm{d}x$, iii) $\int_{-2}^{2} \frac{1}{\sqrt{4-x^2}}\,\mathrm{d}x$, iv) $\int_0^1 \frac{1}{x^2+x+1}\,\mathrm{d}x$

3.(発展課題）次の2重積分を求めよ.

$\iint_D \sqrt{x^2+y^2}\,\mathrm{d}x\,\mathrm{d}y$ D : $0 \le y \le x \le 1$

解答例

1.

```
> int(4*x+3,x);
  int( 1/(1+exp(x)),x);
  int(1/(exp(-x)+exp(x)),x);
  int(sqrt(1-x^2),x);
```

$$2x^2+3x$$

$$-\ln(1+e^x)+\ln(e^x)$$

$$\arctan(e^x)$$

$$\frac{1}{2}x\sqrt{1-x^2}+\frac{1}{2}\arcsin(x) \tag{2.3.1.1}$$

2.

```
> int(sin(x),x=0..Pi);
  int(arctan(x),x=0..1);
  int(1/(sqrt(4-x^(2))),x=-2..2);
  int(1/(x^2+x+1),x=0..1);
```

$$2$$

$$\frac{1}{4}\pi-\frac{1}{2}\ln(2)$$

$$\pi$$

$$\frac{1}{9}\pi\sqrt{3} \tag{2.3.2.1}$$

3.

```
> with(plots):
> inequal({x-y>=0,x>=0,x<=1,y>=0},x=-0.5..1.5,y=-0.5..1.5,
  optionsexcluded=(color=white));
```




```
> f:=unapply(sqrt(x^2+y^2),(x,y)):
> plot3d(f(x,y),x=0..1,y=0..1,axes=box);
```




```
> int(int(f(x,y),y=0..x),x=0..1);
```

$$\frac{1}{6}\sqrt{2}+\frac{1}{6}\ln\left(1+\sqrt{2}\right) \tag{2.3.3.1}$$